# Maintainer: Jan Alexander Steffens (heftig) <heftig@archlinux.org>
# Contributor: Ionut Biru <ibiru@archlinux.org>

pkgname=libwnck3
pkgver=40.0
pkgrel=1
pkgdesc="Library to manage X windows and workspaces (via pagers, tasklists, etc.)"
url="https://gitlab.gnome.org/GNOME/libwnck"
arch=(x86_64)
license=(LGPL)
depends=(gtk3 startup-notification libxres)
makedepends=(gobject-introspection gtk-doc git meson)
_commit=d161ed31914711b97a1c6ebef0558d80ce51113b  # tags/40.0^0
source=("git+https://gitlab.gnome.org/GNOME/libwnck.git#commit=$_commit")
sha256sums=('SKIP')

pkgver() {
    cd libwnck
    git describe --tags | sed 's/-/+/g'
}

prepare() {
    cd libwnck
}

build() {
    arch-meson libwnck build -D gtk_doc=true
    meson compile -C build
}

check() {
    meson test -C build --print-errorlogs
}

package() {
    meson install -C build --destdir "$pkgdir"
}

# vim:set sw=4 et:
